NOAA Technical Memorandum ERL WPL-25

DEPARTMENT OF COMMERCE ★ UNITED STATES OF AMERICA ★

STELLAR SCINTILLOMETER MODEL II FOR MEASUREMENT OF REFRACTIVE-TURBULENCE PROFILES

G. R. Ochs
Ting-i Wang
F. Merrem

Wave Propagation Laboratory
Boulder, Colorado
April 1977

noaa NATIONAL OCEANIC AND ATMOSPHERIC ADMINISTRATION / Environmental Research Laboratories

NOAA Technical Memorandum ERL WPL-25

STELLAR SCINTILLOMETER MODEL II FOR MEASUREMENT OF REFRACTIVE-TURBULENCE PROFILES

G. R. Ochs
Ting-i Wang
F. Merrem

The primary support for this project was provided by RADC(OCSE), Griffiss AFB, New York 13441, under contract F30602-76-0006. The Scientific Monitor was Lt. James Spencer.

Wave Propagation Laboratory
Boulder, Colorado
April 1977

UNITED STATES
DEPARTMENT OF COMMERCE
Juanita M. Kreps, Secretary

NATIONAL OCEANIC AND
ATMOSPHERIC ADMINISTRATION
Robert M. White, Administrator

Environmental Research
Laboratories
Wilmot N. Hess, Director

NOTICE

The Environmental Research Laboratories do not approve, recommend, or endorse any proprietary product or proprietary material mentioned in this publication. No reference shall be made to the Environmental Research Laboratories or to this publication furnished by the Environmental Research Laboratories in any advertising or sales promotion which would indicate or imply that the Environmental Research Laboratories approve, recommend, or endorse any proprietary product or proprietary material mentioned herein, or which has as its purpose an intent to cause directly or indirectly the advertised product to be used or purchased because of this Environmental Research Laboratories publication.

TABLE OF CONTENTS

STELLAR SCINTILLOMETER MODEL II FOR MEASUREMENT OF REFRACTIVE-TURBULENCE PROFILES

G. R. Ochs, Ting-i Wang, and F. Merrem

An optical system for measuring refractive-turbulence profiles in the atmosphere is described. The instrument measures the profile along the light path to a star by analyzing the scintillation of the star by the atmosphere, and is an improved version of an earlier system. The circuit diagram, computer program, and operating instructions for the instrument are included.

1. INTRODUCTION

By properly analyzing the scintillation pattern from a single star, the refractive-turbulence profile along the line-of-sight may be obtained. The instrument here described is capable of measuring C_n^2, the refractive-index structure parameter, in about four independent height regions in the atmosphere within a 5-min observing period. A 36-cm Schmidt-Cassegrain telescope focusses starlight into an attached instrument package which sequentially measures the scintillation intensity present at different spatial wavelengths from 5 to 15 cm, and transmits this information to an on-line minicomputer, which processes the data to obtain C_n^2 values for seven ranges from the stellar scintillometer. The telescope and instrument package are shown in Figure 1.

A particular size of detail (i.e., spatial wavelength) in the scintillation pattern arises generally from a particular height region in the atmosphere. A weighting function can be plotted describing the degree of importance of refractive-index turbulence at different heights in creating scintillation of a certain spatial wavelength on the ground. For example, weighting functions for the finite two-dimensional linear array used in the stellar scintillometer having spatial wavelengths of 5, 10, and 15 cm are shown in Figure 2. By linearly combining weighting functions for three different spatial wavelengths from 5 to 15 cm, it is possible to confine the



Figure 1. The telescope and attached instrument package.



Figure 2. Weighting function of a finite two-dimensional array of detectors observing stellar scintillations for various spatial wavelengths d. The broad band effect of the light source has been included.



Figure 3. Composite path-weighting function by linearly combining the weighting functions at 3 different spatial wavelengths.

turbulence measurement to seven different height regions in the atmosphere, as shown in Figure 3.

2. DESCRIPTION OF THE INSTRUMENT

The instrument package (shown with the cover removed in Figure 4 and schematically in Figure 5) contains the optical spatial filter and the electronics. The optical spatial filter is formed by placing a reticle, consisting of a checkerboard of aluminized squares 0.5 mm on a side, outside the focus of the telescope. The starlight passing between the squares is collected by one PM tube while that reflected from the squares is directed to the other tube. The reticle is angled slightly with respect to the telescope optical axis so that the reflected light may be directed to the second tube by an off-axis mirror. By subtracting the electical output of one tube from that of the other, we form a two-dimensional spatial filter that is sensitive to only one band of spatial wavelengths in the scintillation pattern incident upon the telescope aperture.

The wavelength d sensed by the spatial filter is d=d'F/f, where d' is the spatial wavelength of the reticle, F is the focal length of the telescope, and f is the telescope focus-to-reticle distance. This wavelength is changed by varying f continuously with the motor-driven reticle mount. The electrical output of the filter is a fluctuating signal whose frequency content is determined by scintillation pattern movement and is also sensitive to the orientation of the filter. While the frequency content of the signal will vary both with wind speed, and direction with respect to the checkerboard filter, we are interested only in the amplitude of the signal. The frequency content will not affect the result as long as it is within the electrical passband of the system.



Figure 4. Interior of the instrument package.



Figure 5. Schematic diagram of the optical spatial filter and the electronics for the C_n^2 profiling system.

We wish to measure σ_{I_f} which, in terms of measured quantities, is

$$\sigma_{I_f}^{\;2} = \left[\overline{\left(\frac{I_1'}{I_1} - \frac{I_2'}{I_2}\right)^2}\right]^{\frac{1}{2}} \tag{1}$$

Here I_1 and I_1' are the mean and fluctuating part of the signal from PM tube 1 and I_2 and I_2' are the same for PM tube 2. For $I_1'/I_1 \ll 1, I_2'/I_2 \ll 1$, and with high pass filtering of the difference signal, it can be shown that

$$\sigma_{I_f} \approx \left\{\overline{[\ell n(I_1+I_1') - \ell n(I_2+I_2')]^2}\right\}^{\frac{1}{2}} \tag{2}$$

with negligible error. We choose to measure σ_{I_f} by the differencing method described in Eq. (2) because the differencing procedure improves dynamic range and stability. It is particularly important that the circuit be insensitive to gain differences of the optical and electronic circuits between PM tube 1 and PM tube 2, since a gain balance cannot be maintained in all positions as the reticle translates. We obtain a voltage proportional to σ_{I_f} by using analog circuitry (Fig. 5) to take the difference of the logarithms of signals 1 and 2. We bandpass-filter(6-900 Hz) that voltage and obtain the 1-sec root-mean-square average. The measured σ_{I_f} is then insensitive to both the gain of the system and the mean intensity of the star light. The bandpass ratio is a compromise between wind speed range and signal-to-noise ratio, and the range chosen covers atmospheric wind speeds from approximately 0.9 to 45 m/s. The low frequency cutoff is also determined by the necessity of removing any signal fluctuations due to reticle movement.

We compute σ_I, the log-irradiance standard deviation of the whole aperture, by taking the logarithm of the sum of PM tube signals 1 and 2, bandpass filtering (6-900 Hz), and obtaining the 1-sec root-mean-square average.

The instrument package supplies to the minicomputer four signal voltages labeled F, W, A. and I. They are proportional, respectively, to the following:

σ_{I_f}, the fractional standard deviation of the spatially filtered signal.

d , the wavelength of the spatial filter.

σ_I , the log-irradiance standard deviation of the whole telescope aperture, and

I , the irradiance of the whole telescope aperture.

From the first three inputs we calculate a C_n^2 profile according to the theory presented earlier. The fourth input (I) is used to set signal levels, monitor the guiding of the telescope, and provide a record if clouds obscure the star. The log-irradiance standard deviation of the whole aperture σ_I is the weighted measurement of the refractive-index turbulence which approximately covers the region over which the instrument profiles C_n^2. The profile measurement is calibrated by comparing, for each run, this integrated C_n^2 measurement with the value obtained from a linear combination of weighting functions having nearly the same form as the whole aperture weighting function.

Samples of σ_{I_f}, d, σ_I, and I are digitized at .05 sec intervals. From this information, the computer forms 0.2 sec averages. In order to suppress the measurement error, we use a triangular smoothing window (with halfwidth= 2 cm) centered at the required spatial wavelength to obtain $\sigma_{I_f}{}^2$ at that spatial wavelength and compute C_n^2 at 7 ranges.

3. ALIGNMENT PROCEDURE

3.1 Optical Alignment and Calibration

An eyepiece is provided on the instrument package to check the calibration of the spatial filter. By rotating the rear diagonal mirror 90° and pointing the telescope at a first magnitude star, one can view the spatial filter pattern directly. To read out spatial wavelength on the computer printout insert the following statement in the computer program: 423 PRINT L1. Then the filter motor can be run manually until the computer printout reads 15 cm. This corresponds to 1.17 volts at W. At this point a template having teeth spaced 5 and 15 cm apart is placed directly over the telescope objective. The shadow pattern of squares (parallel to the side of a square, not diagonally) should agree in wavelength with the template. The 5-cm wavelength is checked in the same manner. If the printout does not agree with the direct measurement, run the filter manually until the computer wavelength reads 15 cm. Then adjust the optical filter to a 15-cm wavelength by comparing the shadow bands to the template while adjusting the focus knob of the telescope. The 5-cm wavelength should now check also (3.74 volts at W). If it does not, proceed as follows:

1. Insert: D(Ø).
2. Run the filter motor until W=1.17 volts.
3. Set the optical filter on the telescope for a 15-cm wavelength by adjusting the telescope focus until the shadow pattern as seen in the eyepiece agrees with the 15-cm template wavelength.
4. Read out $D(Ø)_{15}$.
5. Set the optical filter on the telescope to 5 cm. and read out $D(Ø)_5$. This should correspond to about 3.74 volts at W.

6. As the motor runs from W=1.17 to W=3.74 volts the observed spatial wavelength should continuously decrease. If there is a discontinuity, set W=1.17 volts and adjust the telescope focus knob until a second focus position is found where a 15 cm spatial wavelength pattern is obtained. Then go back to step 3.

7. Calculate

$$B = 0.5D(\emptyset)_5 - 1.5D(\emptyset)_{15}$$
$$A = 5(B + D(\emptyset)_5).$$

8. A and B are the required constants to be used in the C_n^2 profile program in step 300, i.e., 300 DEF FNA(X) = A/(X+B).

The optical alignment of the instrument should be such that all of the transmitted and reflected light from the reticle is collected by the two photomultipliers in all positions of the filter. One way to check this is to put small pieces of paper over the PM tubes and visually observe the patterns while the telescope is pointed at a bright but distant light source.

3.2 Electronic Calibration

Remove the instrument box side panel. There are two circuit boards in the instrument. For access to board 2 underneath, remove the board 1 mounting screws and turn it out of the way. The alignment procedure is as follows:

1. With the PM inputs grounded, set b for zero volts at I.
2. With the telescope pointed at a distant light or a star, alternately ground one PM input and then the other, and adjust a for equal signal output at I.
3. Midscale setting of c is satisfactory in this circuit.
4. Ground pin 3 of op amp 4. Set d for zero volts at XF.
5. Disconnect PM inputs and put approximately -1 volt DC in one input and a sine wave of approximately .1 volt RMS with mean -1.5 volt into the other unit. Measure RMS voltage at pin 1 of the 757P and at output XF. Set e for exactly X20 gain.
6. Ground XF and set g for zero at F.
7. Use the input signals described in step J. Measure RMS voltage at pin 1 of the 757P and the output DC voltage at F. Adjust f for exactly X20 gain.
8. On board 2, with PM inputs grounded, set h for zero at pin 6 of op amp 8.
9. Midscale setting of i is satisfactory.
10. Ground pin 3 of op amp 9. Set j for zero volts at output of op amp 11.

11. Using the same test signals as in step 5, set k for exactly X20 circuit gain from pin 1 of the 757P on board 2 to the output of op am 11.

12. Ground output of op amp 11 and set m for zero at A.

13. Using the same test signals as in step 5, measure RMS voltage at pin 1 of the 757P on board 2 and the DC voltage at output A. Adjust 1 for exactly X20 gain.

4. OPERATING PROCEDURE

Select a star (*not* a planet) of second magnitude or preferably brighter, at an elevation of 45° or higher. Since the instrument reads in range along its line-of-sight, a better profile with less correction will be obtained if the star is near the zenith. Center the star in the telescope finder and set the meter switch on the instrument box to I (signal intensity). Turn on the motor. Slowly increase the PM tube voltage until an on-scale meter reading is obtained. Then adjust the telescope slow motion controls to peak the signal. Readjust the PM tube voltage so that the meter reading (I) is about 2/3 full scale. To avoid excessive PM tube currents, the meter reading should not exceed full scale; otherwise the setting is not critical. Because of noise considerations, one should limit the PM tube voltage to about 900. With this limit the meter reading will be less than 2/3 full scale on second magnitude stars. If the telescope drifts off the star or clouds obscure the star, the computer program stops data taking and prints out "LOW SIGNAL". Low signal is activated if I is less than .2 full scale. It might be desirable to lower this level (step 423) for stars less than first magnitude.

After the proper PM tube voltage levels are chosen, close the dome and illuminate it with the DC light provided, adjusting the brightness so that meter reading I is about the same as it was when looking at the star. Using the same PM voltage as on the star, we now have outputs on A (whole aperture variance-D(1) in program) and F(filtered variance-D(2) in program) which are instrument noise levels, principally from the PM tubes. Now insert the following step in the program:

```
421 PRINT D(1);D(2).
```

Run the program. Then set G(1) equal to the average value of D(1) in step 205, and set G(2) equal to the average of D(2) in step 210. This procedure subtracts noise power from signal power under operating conditions. It should be reasonably constant over a number of nights if the same star is used. The correction is of course more important on a second magnitude star.

It should not be necessary to calibrate the spatial filter unless the focus has been changed. Remove step 421, peak up signal I with the telescope slow-motion controls, and the system is ready to run. Once started, the system will continue to run unless the telescope drifts off the star or clouds obscure it. However the light intensity should be monitored and peaked up as necessary during data printout periods when data is not being accumulated. With accurate polar axis alignment of the telescope, tracking should be good for 20 minutes or more.

The program is set to average four data runs before printing out the result; this will take about 3 minutes. If desired, more data can be accumulated before printing out a result. The number of runs used (one through 6) can be set as desired by changing JØ in step 145.

APPENDIX A. Circuit Diagrams



BOARD 2



4250 OP AMPS

APPENDIX B. Computer Programs

2DSP CN2 PROFILE

A basic language program to calculate a C_n^2 profile from a stellar scintillometer. The program makes use of 3 external routines accessed by "CALL" statements. All "CALL" statements in some way supply control information to the A/D converter.

```
10 REM 7/5/77 MAUI
15 REM CAL BY WHOLE APERTURE
20 REM AVERAGE OVER MANY RUNS
30 PRINT
40 PRINT "YEAR-MONTH-DATE (YYMMDD)";
41 INPUT A1
45 PRINT "LOCAL TIME";
46 INPUT A2
48 A2=A2-40* INT(A2/100)
50 A0=1
72 REM ****************
73 REM INPUT PARAMETERS
74 REM ****************
76 REM COEFFICIENTS OF CN2/(LOG-AMP VARIANCE)
100 C[0]=12.18
101 C[1]=5.66
102 C[2]=9.04
103 C[3]=6.19
104 C[4]=6.34
105 C[5]=7.72
106 C[6]=4.57
107 C[7]=3.24
145 J0=4
150 REM  N1 SAMPLES  PER PRINT
160 N1=1+ INT(450*J0/255)
170 REM HALF WIDTH OF FILTER."
180 H0=.1
190 N0=2
195 PRINT "NOISE LEVELS"
205 G[1]=16
210 G[2]=24
212 PRINT "G(1)";G[1];"   G(2)";G[2];
230 REM CHANNEL C1 THRU C2
240 C1=0
250 C2=3
260 CALL 1,C1,C2
270 REM SAMPLING FREQUENCY C3,C4
280 C3=2
290 C4=5
292 CALL 2,C3,C4
295 DIM D[4],E[3],W[7],L[255],S[255],Z[255]
297 DIM R[7,3],B[7],C[7]
```

```
300 DEF  FNA(X)=7230/(X+18)
308 PRINT
309 PRINT "WHOLE APERTURE CALIBRATION"
310 PRINT "CHANNELS";C1;" THRU";C2
320 PRINT "C=";C3," D=";C4
330 PRINT "AVG";N1;" SAMPLES AND";J0;" RUNS."
340 M1=0
342 J9=0
362 REM
363 REM *********************
364 REM DATA INPUT AND FIND S
365 REM *********************
367 N3=0
368 M=0
369 Q5=0
370 FOR J=0 TO 3
380   E[J]=0
390 NEXT J
400 L=0
415 FOR I=1 TO N1
416   CALL 3,D
417   D[0]=D[0]*4
418   D[1]=D[1]*4
419   D[2]=D[2]*4
420   D[3]=D[3]*4
422   L1= FNA(D[0])
424   IF Q5=1 THEN  GOTO 440
426   IF L1<5 THEN  GOTO 432
428   IF L1>18 THEN  GOTO 432
430   GOTO 416
432   L2=L1
433   Q5=1
435   GOTO 416
440   FOR J=1 TO 2
450     E[J]=E[J]+D[J]*D[J]
452     E[J]=E[J]-G[J]*G[J]
455   NEXT J
470   IF D[3]>102 THEN  GOTO 480
471   PRINT  TAB (20);"******** LOW SIGNAL ********"
472   PRINT "";
473   GOTO 472
480   L=L+L1/N1
495 NEXT I
496 IF L>28 THEN  GOTO 370
497 IF L<5 THEN  GOTO 672
498 IF L>18 THEN  GOTO 672
500 E[1]=1.9771E-8*E[1]/N1
505 E[2]=1.9771E-8*E[2]/N1
512 S=E[2]*L↑(-11/3)
514 REM
515 REM *************
516 REM DATA PRINTOUT
517 REM *************
```

```
550 N3=N3+1
555 M=M+E[1]
560 L[N3]=L
570 S[N3]=S
575 Z[N3]=E[1]
672 IF  ABS(L-L2)<11 THEN  GOTO 370
674 IF L<5 THEN  GOTO 682
675 IF L>18 THEN  GOTO 682
680 GOTO 370
682 J9=J9+1
683 A0=A0+1
684 IF J9<>J0* INT((J9+.5)/J0) THEN  GOTO 369
685 REM
686 REM ***********************
690 REM FIND AND PRINT C(N)↑2'S
692 REM ***********************
695 PRINT
700 PRINT "DATE (YYMMDD)";"     TIME (HHMM)"
705 A3=A2+ INT(A0/2+.5)
707 A3=40* INT(A3/60)+A3
710 IF A3>2400 THEN  GOTO 715
711 PRINT  TAB (3);A1; TAB (18);A3
712 GOTO 720
715 PRINT  TAB (3);A1+1; TAB (18); INT(A3-2400+.5)
720 PRINT "LAST WAVELENGTH =";L
725 A0=A0+3
730 M=M/N3
765 PRINT
766 PRINT "  CODE"; TAB (12);"CN↑2";
768 PRINT  TAB (20);"1E-19"; TAB (28);"1E-18"; TAB (36);"1E-17";
769 PRINT  TAB (44);"1E-16"; TAB (52);"1E-15"; TAB (60);"1E-14"
770 PRINT .01* INT(100000*M);
771 PRINT  TAB (22);"I-------I-------I-------I-------I-------I"
775 RESTOR
780 FOR I=0 TO 7
790   W[I]=0
800   FOR J=1 TO 3
810     READ T1
820     GOSUB 1070
830     READ T2
850     W[I]=W[I]+T2*S1
852   NEXT J
853   IF I<>0 THEN  GOTO 855
854   C9=C[0]*M*3.48E-15/W[0]
855   X0=W[I]*C9/C[I]
856   IF I=0 THEN  GOTO 910
875   IF X0>1E-19 THEN  GOTO 880
876   Y=0
877   GOTO 885
880   Y= LOG(X0*1E+19)/2.30258
885   Y1=22+ INT(8*Y+.5)
888   C8=- LOG( ABSX0)/2.30258
889   B[I]= SGN(X0)*.01* INT(100*C8+.5)
890   PRINT B[I]; TAB (10);
```

```
900   GOSUB 9141
903   IF Y1=22 THEN  GOTO 906
904   PRINT  TAB (22);"I"; TAB (Y1);"*"
905   GOTO 910
906   PRINT  TAB (22);"*"
910 NEXT I
915 PRINT "888888"
916 IF A3>2400 THEN  GOTO 920
917 PRINT A1;",";A3;","
918 GOTO 925
920 PRINT A1+1;","; INT(A3-2400+.5);","
925 PRINT .01* INT(100000*M);",";
930 FOR I=1 TO 7
931   PRINT B[I];",";
935 NEXT I
940 PRINT .01* INT(100*C9*1.25E+9+.5);","
945 PRINT "999999"
950 GOTO 340
952 REM
953 REM ****************************
955 REM DATA OF LINEAR COEFFICIENTS
956 REM ****************************
1000 DATA 15.5, 1, 13.5,-.37, 11, 0
1005 DATA 15, 1, 13,-.8, 10, .095
1010 DATA 13.5, 1, 11,-.72, 8.5, .12
1015 DATA 11.5, 1, 9.5,-.43, 15.5,-.27
1020 DATA 10.5, 1, 8,-.31, 15.5,-.38
1025 DATA 9.2, 1, 15.2,-.62, 5, 0
1030 DATA 7.5, 1, 10.5,-.33, 15,-.44
1035 DATA 6, 1, 9,-.44, 14,-.38
1065 REM
1067 REM ************
1070 REM  FIND S(T1)
1073 REM ************
1080 S2=0
1090 S3=0
1091 H=H0*T1
1092 H1=H+H
1094 IF  ABS(T1-5)<H1 THEN  H1= ABS(T1-5)
1096 IF  ABS(T1-18)<H1 THEN  H1= ABS(T1-18)
1100 FOR I1=1 TO N3
1110   G=1- ABS(T1-L[I1])/(2*H)
1115   IF  ABS(T1-L[I1])>H1 THEN  G=0
1120   IF G<0 THEN  G=0
1130   S2=S2+G*S[I1]
1140   S3=S3+G
1150 NEXT I1
1160 IF S3<>0 THEN  GOTO 1170
1163 S1=0
1165 GOTO 1200
1170 S1=S2/S3
1200 RETURN
```

```
9130 REM
9135 REM *************************
9137 REM   ROUND   RSL   10/24/74
9140 REM *************************
9141 Y0=1
9142 FOR I0=1 TO N0
9143   Y0= INT(Y0*10+.5)
9144 NEXT I0
9145 I0= INT( LOG ABSX0/ LOG10+.001)
9146 X0=X0/10↑I0
9147 PRINT  INT(X0*Y0+.5)/Y0;
9148 IF I0=0 THEN  PRINT "  ";
9149 IF I0=0 THEN  GOTO 9151
9150 PRINT "E";I0;
9151 IF  ABSI0<10 THEN  PRINT " ";
9152 RETURN
9500 REM
9505 REM ****************
9510 REM LIST OF NOTATIONS
9520 REM ****************
9600 REM C1 -- CHANNEL STARTED
9610 REM C2 -- CHANNEL ENDED
9620 REM C3,C4 -- SAMPLING FREQUENCY
9650 REM C() -- COEFFICIENTS OF CN2/(LOG-AMP VARIANCE)
9660 REM D(0) -- POSITION OF SPATIAL FILTER
9670 REM D(1) -- VARIANCE OF THE WHOLE APERTURE
9680 REM D(2) -- VARIANCE OF THE SPATIAL FILTERED SIGNAL
9690 REM D(3) -- RECEIVED INTENSITY OF THE WHOLE APERTURE
9700 REM E(1),E(2) -- AVERAGE OF D()*D() OVER N1 SAMPLES
9710 REM E(3) -- AVERAGE OF D(3) OVER N1 SAMPLES
9720 REM G -- SMOOTHING WINDOW FUNCTION
9725 REM H -- WIDTH OF THE SMOOTHING WINDOW
9730 REM L,L() -- AVERAGE SPATIAL WAVELENGTH OVER N1 SAMPLES
9740 REM L1 -- SPATIAL WAVELENGTH
9750 REM M -- MEAN OF E(1) OVER N3 SAMPLES
9755 REM N0 -- DIGITS OF ROUND-OFF
9760 REM N1 -- SAMPLES PER PRINT9770 REM N3 -- SAMPLES PER RUN
9770 REM N3 -- SAMPLES PER RUN
9780 REM S,S() -- VARIANCE * WAVELENGTH ↑ (-11/3) OVER N1 SAMPLES
9785 REM S1 -- S() AFTER SMOOTHING
9790 REM T1 -- INPUT SPATIAL WAVELENGTH
9795 REM T2 -- LINEAR COEFFICIENT OF THE COMPOSITE WEIGHT
9800 REM W() -- COMPOSITE VARIANCE
9810 REM X0 -- CN ↑ 2
9820 REM Z() -- RENAME OF E(1) FROM 1 TO N3

>
```

TEST OF 2DSP CN2 PROFILE

The basic language program to calculate a C_n^2 profile may be tested by disconnecting A/D inputs F, A, and I at the computer and reconnecting all of them to a +2.00 volt test signal. The telescope instrument package must be turned on with the motor running to obtain a suitable input signal at W. The program will ask for the year, month, date, and time before taking data. It will then print out the year, month, and new date and time for each succeeding profile. If LOW SIGNAL occurs, the program should be restarted since the timing will be affected by the LOW SIGNAL printout time.

Between the 888888 and 999999 the data is encoded in a form suitable for transfer from paper tape to cards. It is interpreted as follows.

Entry	Description
770610	Year, month, and date.
1654	Time.
13.29	Whole aperture log amplitude variance $X10^3$.
16.74 through 13.87	$-Log_{10}C_n^2$ listed in order of ascending altitude. A minus sign in front of the number indicates that the computed C_n^2 value is negative.
2.99	A calibration number near 1 used to check the quality of the data. The data is suspect if this number departs more than a factor of 2 from its average.

```
YEAR-MONTH-DATE (YYMMDD)? 77061651
LOCAL TIME?
STOP  @ 46

>RUN

YEAR-MONTH-DATE (YYMMDD)? 770610
LOCAL TIME? 1651
NOISE LEVELS
G(1) 34   G(2) 58
WHOLE APERTURE CALIBRATION
CHANNELS 0 THRU 3
C= 2      D= 5
AVG 8 SAMPLES AND 4 RUNS.

DATE (YYMMDD)      TIME (HHMM)
   770610            1654
LAST WAVELENGTH = 18.0297
```

```
 CODE        CN↑2      1E-19    1E-18    1E-17    1E-16    1E-15    1E-14
 13.29                 I--------I--------I--------I--------I--------I
 16.74      1.81E-17   I                   *
 16.45      3.53E-17   I                     *
 16.39      4.07E-17   I                      *
 16.15      7.01E-17   I                        *
 16.02      9.54E-17   I                         *
 14.38      4.16E-15   I                                      *
 13.87      1.34E-14   I                                          *
888888
 770610, 1654,
 13.29, 16.74, 16.45, 16.39, 16.15, 16.02, 14.38, 13.87, 2.99,
999999
```

BASIC LANGUAGE A/D TEST TAPE

A Basic language program to provide the user with a method of testing or adjusting the A/D converter. The number of channels to be digitized, the sampling frequency and a calibrated voltage reference must be provided by the user. Additional data for the adjusting of the A/D converter can be

found in <u>FIELD ALIGNMENT PROCEDURE</u> (4120-4183), <u>ANALOG TO DIGITAL CONVERTER DIAGNOSTIC</u>, Technical Reference, <u>THE ANALOG DATA CONVERSION SYSTEM</u> and drawings 000 362 (A/D CONVERTER SYSTEM, 000 384 (DIFFERENTIAL BUFFER AMPLIFIER OPTION) and 000 375 (ANALOG MULTIPLEXER PADDLE BOARD.)

Listing of Basic Language A/D Test Tape

```
10 PRINT " CHANNEL CS THRU CHANNEL CE ";
20 INPUT C1,C2
30 CALL 1,C1,C2
40 IF C2-C1<0 THEN  GOTO 10
50 PRINT " SAMPLING FREQUENCY (S/D) "
60 PRINT " S "
70 PRINT " 0 --    60  SAMPLES/SECOND "
80 PRINT " 1 --    10  SAMPLES/SECOND "
90 PRINT " 2 --   100  SAMPLES/SECOND "
100 PRINT " 3 --  1000  SAMPLES/SECOND "
110 PRINT " D -- DIVISOR ";
120 INPUT C3,C4
130 CALL 2,C3,C4
140 DIM D[16]
150 CALL 3,D
160 FOR I=1 TO C2-C1+1
170   PRINT D[I];" ";
180 NEXT I
190 PRINT
200 GOTO 150
210 END

>
```